
I. Introduction

Booster RF stations are divided into two groups, “A” and “B”.  At the beginning of the Booster cycle the phase of the RF applied to A stations are 180 degrees out of phase with the B stations.  The net RF field applied to the beam is zero in this instance, and hence there is no net acceleration or bunching of the beam.  Over the next 600 us interval the phase difference between group “A” and group “B” is driven to zero according to a predefined curve.  

After the initial 600 us interval the phase between group “A” and group “B” is generally fixed except for the addition of an offset at beam “transition” and an offset applied near the end of the booster cycle for the purpose of “bunch rotation”.

This document describes how the generation of the group “A” / group “B” phase control curve is performed by the programmable logic firmware and DSP  processor software loaded onto a Booster VXI DSP Module (BGDSPM).

In this the February 9, 2004 revision a programmable auxiliary offset curve feature and an external analog offset input was added to the paraphase curve and the transition and beam rotation offsets.

II. Specification of the Phase Curve Generation

The curve voltage is generated using an AD7840 DAC on the Booster Generic DSP VXI module.  Timing specifications for the AD7840 states that the DAC can be updated every 90 ns.  The available clock on the BGDSPM is 25 MHz.  We can derive a 833 KHz clock by dividing the 25 MHz clock by 30.  This will give us a DAC update every 1.20 us.  This gives 500 voltage steps within the 600 us interval.  Given that the DAC output is +/- 10 Volts with a 14 bit resolution, the minimum step size is 1.23 mV.  These details are summarized in Table II.1.

Table II.1  Summary of curve generation parameters.

	BGDSPM Clock Frequency
	25 MHz

	DAC Update Clock
	833 KHz

	Time Interval Between DAC Updates
	1.20 us

	Paraphasing Interval
	600 us

	Number of DAC voltage steps
	500

	Minimum DAC Voltage Step Size
	1.23 mV


The phase is shifted from 180 degrees to 0 degrees as the DAC voltage output changes from +10 Volts to 0 Volts.  In the hypothetical case where the transition is a linear ramp each DAC voltage step would be 10 Volts / 500 steps = 20 mV ( phase step = 0.36 degrees).  The voltage error between the desired curve and the discretized output of the DAC can be reduced an order of magnitude with a simple analog filter.

At the end of the 600 us interval the final value of the paraphase curve is held with offsets being added to this value at desired times during the remainder of the Booster cycle.  There are five different offsets that can be summed into the paraphase curve signal.  The first is a base offset for the entire interval.  The next two are single 14 bit signed values each with a programmable time offset parameter that specifies when in the Booster cycle these offsets are applied.  These are the Transition Offset and the Bunch Rotation Offset.  Once these offsets are applied they remain in until the end of the Booster cycle.

The third offset that can be applied if desired, is the auxiliary curve offset.  This is a table of 500 values that can be sequence through and summed into the output curve.  This also has a programmable time offset parameter that specifies when the offset is applied.   Also the clock that sequences through the auxiliary curve offset values in the table is adjustable.  The clock is adjusted by dividing the 833 KHz clock by a value from 1 to 15 (Tclk = 1.2 us to 18 us).  Hence this offset curve can be as long as 9.0 ms.

The final source of offset can be an externally generated analog curve that is digitized by an ADC on the BGDSPM and then summed into the output curve.

Table II.1 gives a list of the parameters for this application that are able to be Read and Written by the ACNET control system. 

Table II.2  List of parameters that can be Read or Written by ACNET.

	Index
	Acnet Name
	Units
	Scale Factor
	Parameter Description

	0
	PARANM
	--
	--
	 Number of values in the base paraphase curve.

	1
	PC1OFF
	Counts
	1.221 Volts per Count
	 C1 Offset: Baseline offset for the paraphase curve.

	2
	PC2OFF
	Counts
	1.221 Volts per Count
	 C2 Offset: Post-Transition offset for the paraphase curve.

	3
	PBROFF
	Counts
	1.221 Volts per Count
	 BR Offset: Beam Rotation offset for the paraphase curve.

	4
	PC2CNT
	Clocks
	1.200 Seconds per Clock
	 C2 Count: Apply Offset C2 at this number of 1.2us clocks from start.

	5
	PBRCNT
	Clocks
	1.200 Seconds per Clock
	 BR Count: Apply Offset BR at this number of 1.2us clocks from start.

	6
	--
	
	
	 Curve 1: The Paraphase Curve.

	7
	--
	
	
	 Curve 2: The Aux Offset Curve.

	8
	PAUXNM
	
	
	 Number of values in the Auxiliary Offset curve.

	9
	PAUXCT
	Clocks
	1.200 Seconds per Clock
	 Aux Curve Count: Apply Aux Offset Curve at this number of 1.2us clocks from the start.

	10
	PAUXDV
	
	
	 Aux Curve Clock Divisor, m.  Clock Period = 1.2us*m.

	11
	PEXTEN
	
	
	 The External Analog Offset enable ( enabled = 1 ).


III. Implementation of the Logic

The logic for the curve player is implemented in an Altera FPGA, part number EPF10K30RC240-4, that sits functionally between the DSP processor and the output DAC’s.  DAC 0 is used as the output of the curve player.

The highest function block in the hierarchy of the curve player is shown in Figure III.1.  A description of its inputs is given in Table III.1.  The essential outputs of the block are described in Table III.2, and the other diagnostic outputs are described in Table III.3.

The curve player block is made up of 8 blocks, itself.  These are the following

Parameter Registers (param_regs).

Curve Player Controller (crv_cntrl_2).

Auxiliary Clock Source (aux_clk).

Address Pointer Sequencer (addr_seq) for both base paraphase and auxiliary curves.

Curve Table Memory (dac_tbl2) for both base paraphase and auxiliary curves.

Offset Summation and Control Block (dac_offsets).

The functions of these blocks are described in the sections to follow.

Figure III.1 Highest hierarchical block of the curve player logic.

Table III.1 Description of the curve player inputs.

	Signal
	I/O
	Description

	LA[13..0]
	I
	These are the Local Address Lines of the BGDSPM Module used in writing the curve player memory and control registers

	ADC_ALTERA_CS/
	I
	This is the overall ADC/DAC Altera PLD chip select.  This chip select is active for all DSP reads and writes to memory locations 0x0042 0000 to 0x0043 FFFF, and direct VXI bus reads and writes to memory at 0x0038 000 to 0x003F FFFF.

	LD[31..16]
	I
	These are the Local Data Lines of the BGDSPM Module used in writing the curve player memory and control registers

	WR/
	I
	This is the DSP processor Write strobe

	DSP_CLK1
	I
	This is the 25 MHz clock for the entire BGDSPM module

	MOD_RESET/
	I
	This is the global logic reset for the entire BGDSPM module

	TCLK_DT1_V
	I
	This is a trigger generated by an external TevClk Decoder module or other external trigger source that starts the playing out of the curve.

	ADC0_[15..2]
	I
	This is the 14 bit value from the output of ADC0.  The analog input of ADC0 is used as the External Analog Offset.


Table III.2 Description of the essential outputs of the curve player block.

	Signal
	I/O
	Description

	PLAY_CLK
	O
	This is the clock that sequences the pointer through the table of values that are loaded into the DAC 0, and acts as the “Load” command for the DAC.  This is the DSP_CLK1 divided by 30.

	PLAY_ENA
	O
	This signal selects the curve player as the source connected to the DAC and enables the curve values onto the DAC 0 input lines.  This signal is active after reset

	SUM_OUT[13..0]
	O
	These are the curve player data output lines that are connected to the DAC 0 input.

	DECODE_PATCH
	O
	This line is used to disable other chip selects coming into the PLD device when the curve player is being addressed.  It is active for all Reads and Writes to addresses 0X0042000F-0X00423FFF


Table III.3 Description of the diagnostic outputs of the curve player block.

	Signal
	I/O
	Description

	SEQUENCE_CLK
	O
	This is the inverse of the PLAY_CLK and is used internal to the curve player to advance the address pointers through the curve table memories.

	CRV1_CNT[15..0]
	O
	This is connected to the register that holds the number of values in the base paraphase curve table.  This value is Written and Read by the DSP.  Only the lower 9 bits are valid.

	CRV2_CNT[15..0]
	O
	This is connected to the register that holds the number of values in the auxiliary curve table.  This value is Written and Read by the DSP.  Only the lower 9 bits are valid.

	C1_OFFSET[15..0]
	O
	This is the value of the initial or base curve offset.  Only the lower 14 bits are valid.  This value is Written and Read by the DSP. 

	C2_OFFSET[15..0]
	O
	This is the value of the beam transition offset.  Only the lower 14 bits are valid.  This value is Written and Read by the DSP. 

	BR_OFFSET[15..0]
	O
	This is the value of the beam rotation offset.  Only the lower 14 bits are valid.  This value is Written and Read by the DSP. 

	Time_Event_1_[15..0]
	O
	This is the number of curve player counts before the C2 offset is applied.  This value is Written and Read by the DSP.

	Time_Event_2_[15..0]
	O
	This is the number of curve player counts before the BR offset is applied.  This value is Written and Read by the DSP.

	Time_Event_3_[15..0]
	O
	This is the number of curve player counts before the auxiliary offset curve is applied.  This value is Written and Read by the DSP.

	AUX_CLK_DIV[15..0]
	O
	This is connected to the  register that holds the divisor for the auxiliary table clock.  Only the lower 4 bits are valid.  This value is Written and Read by the DSP.

	DACTBL_SEL
	O
	This is the chip select that is active when Reading or Writing the base paraphase curve table memory.

	AUXTBL_SEL
	O
	This is the chip select that is active when Reading or Writing the auxiliary curve table memory. 

	CRV1_CNT_CS
	O
	This is the chip select that is active when Reading or Writing the value CRV1_CNT[15..0]. (see above)

	CRV2_CNT_CS
	O
	This is the chip select that is active when Reading or Writing the value CRV2_CNT[15..0]. (see above)

	C1_CS
	O
	This is the chip select that is active when Reading or Writing the value C1_OFFSET[15..0]. (see above)

	C2_CS
	O
	This is the chip select that is active when Reading or Writing the value C2_OFFSET[15..0]. (see above)

	BROT_CS
	O
	This is the chip select that is active when Reading or Writing the value BR_OFFSET[15..0]. (see above)

	C2_OFFSET_CNT_CS
	O
	This is the chip select that is active when Reading or Writing the value Time_Event_1_[15..0]. (see above)

	BR_OFFSET_CNT_CS
	O
	This is the chip select that is active when Reading or Writing the value Time_Event_2_[15..0]. (see above)

	AUX_OFFSET_CNT_CS
	O
	This is the chip select that is active when Reading or Writing the value Time_Event_3_[15..0]. (see above)

	AUX_DIV_SEL
	O
	This is the chip select that is active when Reading or Writing the value AUX_CLK_DIV[15..0] . (see above)

	OVERFLOW
	O
	This bit indicates an overflow in the adder stages used to add in various offsets to the curve player values.

	DAC_TBL[15..0]
	O
	This is the data out of the base paraphase table memory.  Only the upper 14 bits are valid.  This value is summed with the offsets to produce the value that is loaded into DAC 0.  This array of values is Written and Read by the DSP.

	AUX_TBL[15..0]
	O
	This is the data out of the auxiliary offset curve  table memory.  Only the upper 14 bits are valid.  This value is summed with the base paraphase curve to produce the value that is loaded into DAC 0.  This array of values is Written and Read by the DSP.

	TEST
	O
	used in development only

	TEST2
	O
	used in development only


III.1 The Parameter Registers

The Parameter Registers block includes both the flip-flop registers that hold the various parameter values and the address decoder logic for locating the parameters in the memory map of the BGDSPM module.  Given that the ADC_ALTERA_CS/ is active, the address decoder decodes the address lines to derive the signals that either enable Reads and Writes to memory and registers in the curve player, or perform a control function.  Figure III.1.1 shows the Parameter Registers and Decoder block with its inputs and outputs.  Listing III.1.1 shows the Boolean equations used to decode the address lines and indicates where in the DSP memory map the curve player memory and registers appear.  Most of the block’s inputs and outputs were described in the previous section.  Here the START_SEQUENCE command is set active by a pulse at the GP_DIG_IN1 input or a DSP Read or Write to address 0X00420018.  This starts the curve player.  The STOP_SEQUENCE command is set active by a DSP Read or Write to address 0X00420019.  This halts the execution of the curve player, even if it has not completed, and causes the curve player output to reset to the initial point of the curve.  The initial point of the curve is the value that it is expected to start at, at the beginning of the next cycle.

Figure III.1.1  The Parameter Registers and Decoder block.

Listing III.1.1 DAC Table Decoder logic.

DECODE_PATCH = ( !ADC_ALTERA_CS/ & (LA[13..0] >= B"00000000010000") ) ;

DACTBL      =  ( !ADC_ALTERA_CS/ & (LA[13..12] == B"10") ) ;% 0X00422000-0X00422FFF %

AUXTBL      =  ( !ADC_ALTERA_CS/ & (LA[13..12] == B"11") ) ;% 0X00423000-0X00423FFF %

PLAY_ENA    =  ( !ADC_ALTERA_CS/ & (LA[13..0] == B"00000000010110") ) ; % 0X00420016 %

WR_CRV1_CNT =  ( !ADC_ALTERA_CS/ & (LA[13..0] == B"00000000010111") ) ; % 0X00420017 %

WR_CRV2_CNT =  ( !ADC_ALTERA_CS/ & (LA[13..0] == B"00000000011101") ) ; % 0X0042001D %

START_SEQ   =  ( !ADC_ALTERA_CS/ & (LA[13..0] == B"00000000011000") ) ; % 0X00420018 %

STOP_SEQ    =  ( !ADC_ALTERA_CS/ & (LA[13..0] == B"00000000011001") ) ; % 0X00420019 %

WR_C1_OFFSET = ( !ADC_ALTERA_CS/ & (LA[13..0] == B"00000000011010") ) ; % 0X0042001A %

WR_C2_OFFSET = ( !ADC_ALTERA_CS/ & (LA[13..0] == B"00000000011011") ) ; % 0X0042001B %

WR_BR_OFFSET = ( !ADC_ALTERA_CS/ & (LA[13..0] == B"00000000011100") ) ; % 0X0042001C %

WR_TRG1_CNT  = ( !ADC_ALTERA_CS/ & (LA[13..0] == B"00000000100000") ) ; % 0X00420020 %

WR_TRG2_CNT  = ( !ADC_ALTERA_CS/ & (LA[13..0] == B"00000000011110") ) ; % 0X0042001E %

WR_TRG3_CNT  = ( !ADC_ALTERA_CS/ & (LA[13..0] == B"00000000011111") ) ; % 0X0042001F %

ENA_EXT_OFFSET = ( !ADC_ALTERA_CS/ & (LA[13..0] == B"00000000100001") ) ; % 0X00420021 %

AUX_CLK_DIV    = ( !ADC_ALTERA_CS/ & (LA[13..0] == B"00000000100010") ) ; % 0X00420022 %

III.2 The Curve Player Controller

When the curve player runs, the value to be written to the DAC is selected by a 9 bit address pointer into the curve table memory.  Using a binary counter, the Address Pointer Sequencer steps the pointer from address zero to the specified curve count value.  The Curve Player Controller contains three functions.  There is a state machine to time/synchronize the starting and stopping of the pointer sequencing.  There is a block which generates the 833 KHz sequence clock.  There is an event timer that counts the sequence clocks and compares the count to the values in the Time Event Registers and generates a trigger when the count exceeds the register value.  The Curve Player Controller block is shown in Figure III.2.1.

III.2.1 The Player State Machine

The Player State Machine controls the operation of the Address Sequence Pointer counter, the resetting of the Event Timer, and the synchronizing of the Player Clock in response to the START_SEQUENCE and STOP_SEQUENCE commands.  The AHDL file implementing this state machine is given in Listing III.2.1.1.

When the START_SEQUENCE command is received, the RUN_SEQ signal is set active..  Also the Event Timer counter is cleared, and the Player Clock is synchronized.  All of this occurs for a single DSP clock period after the rising edge of START_SEQUENCE.  In the next DSP clock the Address Sequence Pointer counter is enabled.

When the STOP_SEQUENCE command is received the Event Counter is cleared causing all but the initial curve offset to be removed, and the Address Sequence Pointer counter is cleared to zero.  

Figure III.2.1 The Curve Player Controller.

Listing III.2.1.1 AHDL implementation of the curve player state machine.

SUBDESIGN play_states

(

   start, clk, stop, clrn, ready : INPUT;

   ena_cnt, init_ptr, last_ld, sync_clk, clr_trg : OUTPUT;

)

VARIABLE

sm :   MACHINE 

          OF BITS (q1, q2)

          WITH STATES ( 

            rdy  = B"00",

            set  = B"01",


      go   = B"10",

            quit = B"11");

BEGIN


sm.clk = clk;


sm.reset = !clrn;


   TABLE

%  current     current        next                 current                %

%  state       inputs         state                outputs                %

sm  , start, stop, ready => sm , clr_trg, sync_clk, init_ptr, ena_cnt, last_ld;

rdy ,   0  ,   x ,   x   => rdy ,    1   ,     1   ,   1   ,    0   ,     0   ;

rdy ,   1  ,   0 ,   x   => set ,    1   ,     1   ,   1   ,    0   ,     0   ;

rdy ,   1  ,   1 ,   x   => rdy ,    1   ,     1   ,   1   ,    0   ,     0   ;

set ,   x  ,   x ,   x   => go  ,    1   ,     1   ,   1   ,    0   ,     0   ;

go  ,   0  ,   0 ,   x   => go  ,    0   ,     0   ,   0   ,    1   ,     0   ;

go  ,   0  ,   1 ,   x   => quit,    0   ,     0   ,   0   ,    1   ,     0   ;

go  ,   1  ,   x ,   x   => set ,    0   ,     0   ,   0   ,    1   ,     0   ;

quit,   0  ,   0 ,   0   => quit,    1   ,     0   ,   1   ,    0   ,     1   ;

quit,   0  ,   0 ,   1   => rdy ,    1   ,     0   ,   1   ,    0   ,     1   ;

quit,   0  ,   1 ,   x   => quit,    1   ,     0   ,   1   ,    0   ,     1   ;

quit,   1  ,   x ,   x   => set ,    1   ,     0   ,   1   ,    0   ,     1   ;

END TABLE;

END;
III.2.2 The Player Clock

The logic that derives the PLAY_CLK from DSP_CLK1 is shown in Listing III.2.2.1.  This counter divides the DSP clock by 30 resulting in a 1.2 us clock.  The sync signal is used to ensure that the first rising edge of the Address Sequence Pointer counter clock always occurs 600 ns  after the rising edge of the START_SEQUENCE command (to within one DSP clock).  This moves the pointer to the next curve value in the memory.  The play_clk signal, which is the inverse of the sequence_clk, loads the new curve memory word into the DAC on its rising edge.  Therefore, the second curve value in the sequence is applied to the output of the DAC 1.2 us after the START_SEQUENCE command, and a new value is applied every 1.2 us after that. 

Listing III.2.2.1  Derivation of the Play Clock and Sequence Clock.

SUBDESIGN play_clk2

(

   clk, sync : INPUT;

   play_clk, sequence_clk, count[3..0] : OUTPUT;

)

% Clock expected to be (25 MHz / 30) %

VARIABLE

   
count[3..0]    : DFF;


load           : DFF;


toggle

   : TFF;


tgl


   : NODE;

BEGIN

   
count[].clk = clk;


load.clk = clk;


load.d   =  tgl;


toggle.t   = VCC;


toggle.clk = load.q;


toggle.clrn = !sync;

   
IF sync THEN

      count[].d = B"1011";

   
ELSIF load.q THEN

      count[].d = B"1101";


ELSE 

      count[].d = count[].q - 1;

   
END IF;


IF (count[].q == B"0000") THEN tgl = VCC; ELSE tgl = GND; END IF;

   
sequence_clk = toggle.q;


play_clk = !sequence_clk;

END;
III.2.3 The Event Timer

The Event Timer is a counter that starts at zero and is clocked by the SEQUENCE_CLK.  The output of the counter is compared to the values stored in the Timer_Event_1 and Timer_Event_2 registers that were Written by the DSP.  The event bits out of the Event Timer become active once the counter exceeds the value in its associated event register.  The counter is reset by the START_SEQUENCE and STOP_SEQUENCE commands.  The count will stop on its own once it reaches 32768 ( = 39 ms).  Listing III.2.3.1 gives the AHDL implementation of the Event Timer block.

Listing III.2.3.1  AHDL implementation of the Event Timer block.

SUBDESIGN eventtimer

(

   clk, clrn : INPUT;

   event1[15..0], event2[15..0], event3[15..0] : INPUT;

   trg1, trg2, trg3 : OUTPUT;

)

% Clock expected to be (25 MHz / 32) %

VARIABLE

   
count[15..0]    : DFF;


ena

   : NODE;

BEGIN

   count[].clk = clk;

   count[].clrn = clrn;

   IF (ena) THEN

      count[].d = count[].q + 1;

   ELSE

      count[].d = count[].q;

   END IF;


IF (count[] >= event1[]) THEN trg1 = VCC; ELSE trg1 = GND; END IF;

  
IF (count[] >= event2[]) THEN trg2 = VCC; ELSE trg2 = GND; END IF;

  
IF (count[] >= event3[]) THEN trg3 = VCC; ELSE trg3 = GND; END IF;


% Stop the count at 32768 (approx. 39 ms) %

  
IF (count[] >= H"8000") THEN ena = GND; ELSE ena = VCC; END IF;

END;

III.4  The Auxiliary Clock Source

This block produces the pointer sequence clock for playing out the Auxiliary Curve from the Auxiliary Curve Table memory.  This clock can be divided down by the programmable Auxiliary Clock Divisor.  This clock is identical to the SEQUENCE_CLK when the clock divider is set to 1 or zero.  The clock does not start until the AUX_CRV_TRIG signal (Timer Event 3) becomes active.  The AHDL code for this block is given in Listing III.4.1

Listing III.4.1  The Auxiliary Clock Source AHDL Code.

SUBDESIGN aux_clk

(

   clk, enable, clrn, m[3..0] : INPUT;

   aux_clk, count[3..0] : OUTPUT;

)

% Clock expected to be (25 MHz / 30) %

VARIABLE

   
count[3..0]    : DFFE;


match          : DFF;


tst


   : NODE;


ready/

   : NODE;

BEGIN


ready/ = clrn & enable;

   
count[].clk = clk;


count[3..1].clrn = ready/;


count[0].prn     = ready/; % Initialize counter to 0001 %


count[].ena = enable;


match.clk = !clk;


match.d   =  tst;


match.clrn = ready/; % Ensure this reg is Low before things start. %

   
IF match.q THEN count[].d = B"0001";



    ELSE count[].d = count[].q + 1;

   
END IF;


IF (count[].q == m[]) THEN tst = VCC; 




      ELSE tst = GND; 


END IF;

   
IF (m[] <= B"0001") THEN aux_clk = clk; 




     ELSE aux_clk = match.q; 


END IF;

END;

III.5  The Address Pointer Sequencer

The Address Pointer Sequencer is a counter clocked by the Sequence Clock, whose output is used as a pointer into the curve table memory.  The counter, and hence the memory pointer, starts at zero and increments until the value is equal to the set number of curve points.  Listing III.5.1 gives the AHDL code for this block.

Listing III.5.1  AHDL code for the Address Sequencer.

SUBDESIGN addr_seq

(

   seq_clk, player_ena, run_seq, clrn, init_ptr, stop_cnt[8..0] : INPUT;

   cout[8..0] : OUTPUT;

)

VARIABLE

   count[8..0]    : DFF;

   reset_seq      :NODE;

   ena
         :NODE;

BEGIN

   reset_seq = !clrn # init_ptr;

   count[].clk = seq_clk;

   count[].clrn = !reset_seq;

   IF (ena) THEN

      count[].d = count[].q + 1;

   ELSE

      count[].d = count[].q;

   END IF;

   IF (count[] >= stop_cnt[]) THEN ena = GND; 

   ELSE ena = player_ena & run_seq; 

   END IF;

   cout[] = count[].q;

END;

III.6  The Offset Summation and Control Block

The Offset Summation and Control Block contains five 14 bit signed adders that add the five offsets to the base paraphase curve. at appropriate times during the Booster cycle.  The five offsets are

1. The Base Offset:  This offset is added to the DAC Table value through out the Booster cycle.  It compensates for offsets in the Phase Shifter Module which takes the analog DAC voltage as input and provides phase shifted RF reference voltages for Group A and Group B.

2. The Post-Transition Offset:  This offset is added in after transition in the Booster cycle to compensate for changes in the beam response after transition.

3. The Bunch Rotation Offset:  This is a relatively large offset needed to perform bunch rotation.  This occurs near the end of the Booster cycle when it is used.

4. The Auxiliary Offset Curve:  This additional table of points is included for flexibility and provides the ability to adjust the phase curve temporarily when experimenting with the phase control.

5. The External Analog Offset:  This allows the ability to use other curve generating devices in conjunction with the BGDSPM paraphase curve source.

The Offset Summation and Control block is shown in Figure III.6.1.  A description of its inputs and outputs is summarized in Table III.6.1.

[image: image4.png]DSP_CLK1
T DAC_TBL[13.0] | | CLK SUM_OUT[13..0]

— |
TT_OFFSETS.0] | DATALNS.0] OVERFLOW

EXT_CRVZ_ 1130 | ¢ 1130
~AUX_CRVITS.0] | EXT_-CRVI13.0]

T2 OFFSETIS.0] | AUX_CRVI13.0]
[l L B
BR_OFFSETS.0 | c2-13.01

BROT_[13..0]

EXT_CRV_ENA
AUX_CRV_TRIG | | ENA_EXT_CRV
C2_TRIGGER ENA_AUX_CRV
BR_TRIGGER ENA_C2
ENA_BROT

inst17




Figure III.6.1  The Offset Summation and Control block.

Table III.6.1 The Offset Summation and Control block I/O.

	Signal
	I/O
	Description

	DSP_CLK1
	I
	This is the 25 MHz clock for the entire BGDSPM module.

	DAC_TBL_[13..0]
	I
	These are the Base Paraphase Curve values from table memory.

	C1_OFFSET[13..0]
	I
	This is the Base Offset value.

	EXT_CRV2_[13..0]
	I
	This is the External Analog Offset value from ADC0.

	AUX_CRV[13..0]
	I
	These are the Auxiliary Offset Curve values from table memory.

	C2_OFFSET[13..0]
	I
	This is the Post Transition Offset value.

	BR_OFFSET[13..0]
	I
	This is the Bunch Rotation Offset value.

	EXT_CRV_ENA
	I
	This signal gates on the External Analog Offset

	AUX_CRV_ENA
	I
	This signal gates on the Auxiliary Curve Offset.

	C2_TRIGGER
	I
	This signal gates on the Post Transition Offset.

	BR_TRIGGER
	I
	This signal gates on the Bunch Rotation Offset.

	SUM_OUT[13..0]
	O
	These are the curve player data output lines that are connected to the DAC 0 input.

	OVERFLOW
	O
	This bit indicates an overflow in the adder stages used to add in various offsets to the curve player values.

Figure III.6.2  Offset Summation and Control block, Control.

Figure III.6.3  Offset Summation and Control block, Summation.

IV.  Signal Routing on the BGDSPM Module

The Booster Generic Digital Signal Processing Module provides for 3 analog inputs, 3 analog outputs, and many digital inputs and outputs.  The Paraphase application is relatively simple.  It uses only 1 digital input for the start of curve trigger, 1 analog input for the external bias input, and 1 analog output that is the paraphase curve.

The start trigger input is labeled on the front panel.  Figure IV.1 maps out the routing of this trigger from the Lemo connection on the front panel to the Start Trigger input of the curve generator logic and the Interrupt Request input of the DSP processor.

Figure IV.1  Routing of the trigger input.

